
XML and Semantic Web Technologies

XML and Semantic Web Technologies

II. XML / 6. XML Query Language (XQuery)

Lars Schmidt-Thieme

Information Systems and Machine Learning Lab (ISMLL)

Institute of Economics and Information Systems

& Institute of Computer Science

University of Hildesheim

http://www.ismll.uni-hildesheim.de

Lars Schmidt-Thieme, Information Systems and Machine Learning Lab (ISMLL), University of Hildesheim, Germany,

Course on XML and Semantic Web Technologies, summer term 2009 1/33

XML and Semantic Web Technologies

II. XML / 6. XML Query Language (XQuery)

1. Creating Nodes of the Result Tree

2. FLWOR expressions

3. Variables and Functions

4. Types and Static Type Checking

Lars Schmidt-Thieme, Information Systems and Machine Learning Lab (ISMLL), University of Hildesheim, Germany,

Course on XML and Semantic Web Technologies, summer term 2009 1/33



XML and Semantic Web Technologies / 1. Creating Nodes of the Result Tree

XQuery Specification

XQuery is specified in

1. XQuery 1.0: An XML Query Language (Rec 2007/01/23) and

2. XML Syntax for XQuery 1.0 (XQueryX; (Rec 2007/01/23)

as well as documents about requirements, use cases, serialization, and formal
semantics.

XQuery extends XPath 2.0, i.e., (most) any XPath expressions are XQuery "queries".

XQuery does not have an XML Syntax
(like XPath, but contrary to XSLT).

XQuery Modules

The XQuery processing unit is the module:

〈Module〉 := ( xquery version 〈StringLiteral〉 ; )?
( 〈ModuleDecl〉 ; )?
〈Prolog〉
〈Expr〉

Usually one module is stored in one file.

Library modules have a module declaration, but no body expression;
main modules have a body expression, but no module declaration.

All XPath expressions are XQuery expressions 〈Expr〉.

1 xquery version "1.0" ;
2 //title[contains(string(.),"XML")]/../author

Figure 1: Example XQuery consisting of an XPath expression.

Literal Result Nodes

Like in XSLT, nodes can be created in two ways:

1. literal result elements (called direct constructors in the XQuery spec), their
attributes, namespaces and nested character data is copied literally to the result
tree.

XML comments and PIs can be created in the same way (as XQuery does not
have an XML syntax).

1 xquery version "1.0" ;
2 <article author="John Doe" version="2004/06/07">
3 <title>What <em>others</em> say</title>
4 A <em>short</em><!-- 20 pages--> overview ...
5 </article>

Figure 2: Example XQuery consisting of literal result items only.

Literal Result Node Sequences

1 xquery version "1.0" ;
2 <!-- first ideas -->,
3 <?xml-stylesheet href=’article.css’ type=’text/css’?>,
4 <article author="John Doe" version="2004/06/07">
5 <title>What <em>others</em> say</title>
6 A <em>short</em><!-- 20 pages--> overview ...
7 </article>

Figure 3: Example XQuery consisting of literal result items almost only.

Computed Result Nodes

2. computed result nodes (called computed constructors in the XQuery spec):

〈computedConstructor〉 := document { 〈Expr〉? }

| element 〈QName〉 { 〈Expr〉? }

| attribute 〈QName〉 { 〈Expr〉? }

| text { 〈Expr〉? }

| comment { 〈Expr〉? }

| processing-instruction 〈NCName〉 { 〈Expr〉? }

| namespace 〈NCName〉 { 〈Expr〉? }

〈QName〉 and 〈NCName〉 can also be replaced by { 〈Expr〉 }.

Computed Result Nodes

1 <?xml version="1.1"?>
2 <!-- first ideas -->
3 <?xml-stylesheet href=’article.css’ type=’text/css’?>
4 <article author="John Doe" version="2004/06/07">
5 <title>What <em>others</em> say</title>
6 A <em>short</em><!-- 20 pages--> overview ...
7 </article>

Figure 4: Sample XML document.

1 xquery version "1.0" ;
2 for $a in //article return
3 element article {
4 for $att in $a/@* return
5 element { name($att) } { string($att) },
6 element title { string($a/title) }
7 }

Figure 5: XQuery element.xq creating ele-

ments with fixed and computed names.

1 <?xml version="1.0" encoding="UTF-8"?>
2 <article>
3 <author>John Doe</author>
4 <version>2004/06/07</version>
5 <title>What others say</title>
6 </article>

Figure 6: Result document of the query.

Nesting Computed Result Nodes in Literal Result Nodes

In

• values of attributes and

• the contents

of literal result elements, expressions of type

{ 〈Expr〉 }

can be used to insert computed items (for attribute values) or nodes (for element
content).

1 xquery version "1.0" ;
2 for $a in //article return
3 <article>
4 { for $att in $a/@* return
5 element { name($att) } { string($att) },
6 <title>{ string($a/title) }</title>
7 }
8 </article>

Figure 7: XQuery creating elements with fixed and computed names.

If in these contexts "{" or "}" should appear literally, the symbols must be escaped
as "{{" and "}}".

Performing XQuery Queries by Saxon

XQuery queries can be performed, e.g., by Saxon.

call (with saxon8.jar in classpath):

java net.sf.saxon.Query -s anarticle.xml element.xq

Clauses

〈FLWORExpr〉 := (〈ForClause〉 | 〈LetClause〉)+
(where 〈ExprSingle〉)? 〈OrderByClause〉?
return 〈Expr〉

〈ForClause〉 := for

$ 〈QName〉 (as 〈SequenceType〉)? (at $ 〈QName〉)? in 〈Expr〉
(, $ 〈QName〉 (as 〈SequenceType〉)? (at $ 〈QName〉)? in 〈Expr〉)*

〈LetClause〉 := let $ 〈QName〉 (as 〈SequenceType〉)? := 〈Expr〉
(, $ 〈QName〉 (as 〈SequenceType〉)? := 〈Expr〉)*

XPath’s for-expressions is a special case of an XQuery FLWOR expression.
New is:

• let binds additional variables,

• where filters tuples,

• order by orders tuples,

• at $ 〈QName〉 binds an additional positional variable,

• as 〈SequenceType〉 types the for-/let-variable.

for and let Clauses

1 xquery version "1.0" ;
2 for $s in (<one/>, <two/>, <three/>)
3 return <out>{$s}</out>

Figure 8: FLWOR-expression using for.

1 xquery version "1.0" ;
2 let $s := (<one/>, <two/>, <three/>)
3 return <out>{$s}</out>

Figure 9: FLWOR-expression using let.

1 <?xml version="1.0"?>
2 <out>
3 <one/>
4 </out>
5 <out>
6 <two/>
7 </out>
8 <out>
9 <three/>

10 </out>

Figure 10: Result of for-expression.

1 <?xml version="1.0"?>
2 <out>
3 <one/>
4 <two/>
5 <three/>
6 </out>

Figure 11: Result of let-expression.

where Clause

1 xquery version "1.0" ;
2 let $inputvalues := 1 to 1000 return
3 avg(for $x at $i in $inputvalues
4 where $i mod 100 = 0
5 return $x)

Figure 12: FLWOR-expression using where.

1 xquery version "1.0" ;
2 let $inputvalues := 1 to 1000 return
3 avg($inputvalues[position() mod 100 = 0])

Figure 13: Same query using a predicate.

1 550

Figure 14: Result of the queries.

order by Clause

〈OrderByClause〉 := (order by | stable order by) 〈OrderSpecList〉

〈OrderSpecList〉 := 〈OrderSpec〉 (, 〈OrderSpec〉)*

〈OrderSpec〉 := 〈Expr〉 〈OrderModifier〉

〈OrderModifier〉 := ( ascending | descending )?
( empty greatest | empty least )?

1 xquery version "1.0" ;
2 <html><body>
3 Authors: <ol>
4 { for $a in distinct-values(//author)
5 let $sn := substring-after($a, ’ ’),
6 $fn := substring-before($a, ’ ’)
7 order by $sn, $fn
8 return <li>{ concat($sn, ", ", $fn) }</li>
9 }

10 </ol></body></html>

Figure 15: FLWOR-expression with order by

clause.

1 <html>
2 <body>
3 Authors:
4 <ol>
5 <li>Eckstein, Rainer</li>
6 <li>Eckstein, Silke</li>
7 <li>Muellner, Leonard</li>
8 <li>T. Ray, Erik</li>
9 <li>Walsh, Norman</li>

10 </ol>
11 </body>
12 </html>

Figure 16: Result of the query on the

books.xml document.

Positional Variable with at

at allows to bind the position of the actual item to a variable.

This is the only possibility to access the position
(as the XPath function position() retrieves the position of the context item only).

1 xquery version "1.0" ;
2 <books>{
3 for $b at $no in //book return
4 <book no="{$no}">{ string($b/title) }</book>
5 }</books>

Figure 17: FLWOR-expression with positional variable.

1 <?xml version="1.0" encoding="UTF-8"?>
2 <books>
3 <book no="1">XML und Datenmodellierung</book>
4 <book no="2">Learning XML</book>
5 <book no="3">DocBook: The Definitive Guide</book>
6 </books>

Figure 18: Result of the query on the books.xml document.

XPath’s for-expressions and XQuery’s FLWOR-expressions

clause equivalent in XPath
if if
– at position(), if used in where-clause;

none, else
– as cast as, for atomic types;

none, else
let variable can be replaced by its value
where predicates in if-expressions
order by none

Namespaces

〈Prolog〉 := ((〈Setter〉 | 〈ImportDecl〉
| 〈ImportSchemaDecl〉 | 〈NamespaceDecl〉 ; )*

((〈VarDecl〉 | 〈FunctionDecl〉 | 〈OptionDecl〉) ; )*

〈NamespaceDecl〉 := declare namespace 〈NCName〉 = 〈StringLiteral〉
| declare default element namespace 〈StringLiteral〉
| declare default function namespace 〈StringLiteral〉

Variables

〈VarDecl〉 := declare variable $ 〈QName〉 (as 〈SequenceType〉)? ((:=
〈Expr〉 ) | external)

XQuery variables are read-only.

Alternatively, variables can be bound with a let-clause of a FLWOR-expression.

1 xquery version "1.0" ;
2 declare variable $no := 0 ;
3 <books>{
4 for $b in //book
5 let $no := $no + 1 return
6 <book no="{$no}">{ string($b/title) }</book>
7 }</books>

Figure 19: What might happen here?

Lars Schmidt-Thieme, Information Systems and Machine Learning Lab (ISMLL), University of Hildesheim, Germany,

Course on XML and Semantic Web Technologies, summer term 2009 16/33



XML and Semantic Web Technologies / 3. Variables and Functions

Variables

1 <?xml version="1.0" encoding="UTF-8"?>
2 <books>
3 <book no="1">XML und Datenmodellierung</book>
4 <book no="1">Learning XML</book>
5 <book no="1">DocBook: The Definitive Guide</book>
6 </books>

Figure 20: Result of broken query on last slide.

Defining Functions

〈FunctionDecl〉 := declare function 〈QName〉 ( 〈ParamList〉? )

( as 〈SequenceType〉)? ({ 〈Expr〉 } | external)

〈ParamList〉 := 〈Param〉 ( , 〈Param〉 )*

〈Param〉 := $ 〈QName〉 ( as 〈SequenceType〉 )?

1 xquery version "1.0" ;
2 declare namespace f = "http://www.cgnm.de/xml/xquery/functions";
3 declare function f:shortauthor($author as node()*) {
4 let $numAuthors := count($author) return
5 concat($author[1], if ($numAuthors=2) then concat(’ and ’, $author[2])
6 else if ($numAuthors>=3) then ’ et al.’ else ’’)
7 };
8 <html><body><ol>
9 { for $b in //book return

10 <li>{ concat(f:shortauthor($b/author), ": ", string($b/title), ", ", $b/year) }</li>
11 }</ol></body></html>

Figure 21: Function declaration.

For Saxon, names of user-defined functions must have a namespace.

Defining Functions

1 <html>
2 <body>
3 <ol>
4 <li>Rainer Eckstein and Silke Eckstein: XML und Datenmodellierung, 2004</li>
5 <li>Erik T. Ray: Learning XML, 2003</li>
6 <li>Norman Walsh and Leonard Muellner: DocBook: The Definitive Guide, 1999</li>
7 </ol>
8 </body>
9 </html>

Figure 22: Result of query on last slide.

Library Modules

〈ModuleDecl〉 := module namespace 〈NCName〉 = 〈StringLiteral〉

〈ImportDecl〉 := import module (namespace 〈NCName〉 =)? 〈StringLiteral〉
(at 〈StringLiteral〉)?

1 xquery version "1.0" ;
2 module namespace f = "http://www.cgnm.de/xml/xquery/functions";
3 declare function f:shortauthor($author as node()*) {
4 let $numAuthors := count($author) return
5 concat($author[1], if ($numAuthors=2) then concat(’ and ’, $author[2])
6 else if ($numAuthors>=3) then ’ et al.’ else ’’)
7 };

Figure 23: Library module biblib.xq.

1 xquery version "1.0" ;
2 import module namespace f = "http://www.cgnm.de/xml/xquery/functions" at "biblib.xq";
3 <html><body><ol>
4 { for $b in //book return
5 <li>{ concat(f:shortauthor($b/author), ": ", string($b/title), ", ", $b/year) }</li>
6 }</ol></body></html>

Figure 24: Query importing a library module.

Working with Types
To use types defined in a schema, e.g., as

• types of variables,

• as types of function parameters or return types,

• in cast expressions, etc.

the namespace of the schema defining the types has to be declared.

1 xquery version "1.0" ;
2 declare namespace f = "http://www.cgnm.de/xml/xquery/functions";
3 declare namespace xs="http://www.w3.org/2001/XMLSchema";
4

5 declare function f:shortauthor($author as node()*) as xs:string {
6 let $numAuthors := count($author) return
7 concat($author[1], if ($numAuthors=2) then concat(’ and ’, $author[2])
8 else if ($numAuthors>=3) then ’ et al.’ else ’’)
9 };

10 <html><body><ol>
11 { for $b in //book return
12 <li>{ concat(f:shortauthor($b/author), ": ", string($b/title), ", ", $b/year) }</li>
13 }</ol></body></html>

Figure 25: Working with XML Schema types.

Validating and Static Type Checking (1/5)

1 <?xml version="1.0"?>
2 <xs:schema version="1.0" xmlns:xs="http://www.w3.org/2001/XMLSchema"
3 targetNamespace="http://www.cgnm.de/xml/books.xsd"
4 xmlns="http://www.cgnm.de/xml/books.xsd"
5 elementFormDefault="qualified">
6 <xs:element name="books">
7 <xs:complexType>
8 <xs:sequence maxOccurs="unbounded">
9 <xs:element ref="book"/>

10 </xs:sequence>
11 </xs:complexType>
12 </xs:element>
13 <xs:element name="book">
14 <xs:complexType>
15 <xs:sequence>
16 <xs:element name="author" minOccurs="1" maxOccurs="unbounded"
17 type="xs:string"/>
18 <xs:element name="title" type="xs:string"/>
19 <xs:element name="year" type="xs:string"/>

20 </xs:sequence>
21 <xs:attribute name="isbn" type="xs:string"/>
22 <xs:attribute name="cites">
23 <xs:simpleType><xs:list itemType="xs:string"/></xs:simpleType>
24 </xs:attribute>
25 </xs:complexType>
26 </xs:element>
27 </xs:schema>

Figure 26: XML Schema books.xsd of the source document.

1 <?xml version="1.1"?>
2 <books>
3 <book isbn="3-89864-222-4" cites="0-596-00420-6 1-565-92580-7">
4 <author>Rainer Eckstein</author><author>Silke Eckstein</author>
5 <title>XML und Datenmodellierung</title><year>2004</year></book>
6 <book isbn="0-596-00420-6">
7 <author>Erik T. Ray</author><title>Learning XML</title><year>2003</year></book>
8 <book isbn="1-565-92580-7">
9 <author>Norman Walsh</author><author>Leonard Muellner</author>

10 <title>DocBook: The Definitive Guide</title><year>1999</year></book>
11 </books>

Figure 27: Source document books.xml.

Validating and Static Type Checking (2/5)

1 <?xml version="1.0"?>
2 <xs:schema version="1.0" xmlns:xs="http://www.w3.org/2001/XMLSchema"
3 targetNamespace="http://www.cgnm.de/xml/html-toy.xsd"
4 xmlns="http://www.cgnm.de/xml/html-toy.xsd"
5 elementFormDefault="qualified">
6

7 <xs:element name="html">
8 <xs:complexType>
9 <xs:sequence>

10 <xs:element name="body">
11 <xs:complexType mixed="true">
12 <xs:sequence minOccurs="0" maxOccurs="unbounded">
13 <xs:element name="ol">
14 <xs:complexType>
15 <xs:sequence minOccurs="0" maxOccurs="unbounded">
16 <xs:element name="li" type="xs:string"/>
17 </xs:sequence>
18 </xs:complexType>
19 </xs:element>

20 </xs:sequence>
21 </xs:complexType>
22 </xs:element>
23 </xs:sequence>
24 </xs:complexType>
25 </xs:element>
26 </xs:schema>

Figure 28: XML Schema of the target document.

Validating and Static Type Checking (3/5)

1 xquery version "1.0" ;
2 declare namespace f = "http://www.cgnm.de/xml/xquery/functions";
3 declare function f:shortauthor($author as node()*) {
4 let $numAuthors := count($author) return
5 concat($author[1], if ($numAuthors=2) then concat(’ and ’, $author[2])
6 else if ($numAuthors>=3) then ’ et al.’ else ’’)
7 };
8 <html><body><ol>
9 { for $b in //book return

10 <item>{ concat(f:shortauthor($b/auhtor), ": ", string($b/title), ", ", $b/year) }</item>
11 }</ol></body></html>

Figure 29: Query containing errors w.r.t. schemata.

Validating and Static Type Checking (3.5/5)
1 <html>
2 <body>
3 <ol>
4 <item>: XML und Datenmodellierung, 2004</item>
5 <item>: Learning XML, 2003</item>
6 <item>: DocBook: The Definitive Guide, 1999</item>
7 </ol>
8 </body>
9 </html>

Figure 30: Result of broken query on last slide.

Figure 31: Result viewed in HTML browser.

Kinds of Validation

Validity of specific input documents can be checked by the parser.

Validity of specific output documents can be checked by:

1. validating output as a post-process or by,

2. validating output while it is generated.

This is called dynamic type checking (depends on data, done at runtime)
as opposed to static type checking (does not depend on data, done at query
compile time).

Validate Expressions

〈ImportSchemaDecl〉 := import schema 〈SchemaPrefix〉? 〈StringLiteral〉
(at 〈StringLiteral〉)?

〈SchemaPrefix〉 := ( namespace 〈NCName〉 = ) | default element namespace

〈ValidateExpr〉 := validate (lax | strict)? { 〈Expr〉 }

validate returns its argument node with type annotation
(and raises an error, if it is not valid w.r.t. its associated schema).

Validating and Static Type Checking (4/5)

1 xquery version "1.0" ;
2 declare namespace f = "http://www.cgnm.de/xml/xquery/functions";
3 declare function f:shortauthor($author as node()*) {
4 let $numAuthors := count($author) return
5 concat($author[1], if ($numAuthors=2) then concat(’ and ’, $author[2])
6 else if ($numAuthors>=3) then ’ et al.’ else ’’)
7 };
8 import schema namespace bk="http://www.cgnm.de/xml/books.xsd" at "books.xsd";
9 import schema default element namespace "http://www.cgnm.de/xml/html-toy.xsd" at "html-toy.xsd";

10 validate strict {
11 <html><body><ol>
12 { for $b in //bk:book return
13 <item>{ concat(f:shortauthor($b/bk:auhtor), ": ", string($b/bk:title), ", ", $b/bk:year) }</item>
14 }</ol></body></html>
15 }

Figure 32: Query containing errors w.r.t. schemata, with schema import.

Validating and Static Type Checking (5/5)

1 xquery version "1.0" ;
2 declare namespace f = "http://www.cgnm.de/xml/xquery/functions";
3 declare function f:shortauthor($author as node()*) {
4 let $numAuthors := count($author) return
5 concat($author[1], if ($numAuthors=2) then concat(’ and ’, $author[2])
6 else if ($numAuthors>=3) then ’ et al.’ else ’’)
7 };
8 import schema namespace bk="http://www.cgnm.de/xml/books.xsd" at "books.xsd";
9 import schema default element namespace "http://www.cgnm.de/xml/html-toy.xsd" at "html-toy.xsd";

10 validate strict {
11 <html><body><ol>
12 { for $b in //bk:book return
13 <li>{ concat(f:shortauthor($b/bk:author), ": ", string($b/bk:title), ", ", $b/bk:year) }</li>
14 }</ol></body></html>
15 }

Figure 33: Query with errors fixed.

Comparison XSLT–XQuery

a. commonalities:
topic details
0. basic expressions both languages extend XPath.
1. variables read-only
2. definition of functions callable in XPath expressions
3. nested iterations / joins FLWOR-expressions and for-each,

for-each-group

4. creating nodes both languages provide two mechanisms:

• literal result nodes,

• computed result nodes.

5. type strictness both languages build on XML Schema and allow
static type checking

Comparison XSLT–XQuery

b. differences:
topic XSLT XQuery
1. syntax XML non-XML
2. user-defined functions can be used in

expressions and in
value-of

can be used everywhere

3. FLWOR expressions can be replaced by
combined statements
(procedural)

declarative

4. template rules template and
apply-templates

–

5. type strictness maybe not encouraged by
templates

encourgaed by FLWOR
expressions

